asmprobit postestimation — Postestimation tools for asmprobit

Description

The following postestimation commands are of special interest after `asmprobit`:

Command	Description
`estat alternatives`	alternative summary statistics
`estat covariance`	covariance matrix of the latent-variable errors for the alternatives
`estat correlation`	correlation matrix of the latent-variable errors for the alternatives
`estat facweights`	covariance factor weights matrix
`estat mfx`	marginal effects

The following standard postestimation commands are also available:

Command	Description
`estat ic`	Akaike's and Schwarz's Bayesian information criteria (AIC and BIC)
`estat summarize`	summary statistics for the estimation sample
`estat vce`	variance–covariance matrix of the estimators (VCE)
`estimates`	cataloging estimation results
`lincom`	point estimates, standard errors, testing, and inference for linear combinations of coefficients
`lrtest`	likelihood-ratio test
`nlcom`	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
`predict`	predicted probabilities, estimated linear predictor and its standard error
`predictnl`	point estimates, standard errors, testing, and inference for generalized predictions
`test`	Wald tests of simple and composite linear hypotheses
`testnl`	Wald tests of nonlinear hypotheses

Special-interest postestimation commands

`estat alternatives` displays summary statistics about the alternatives in the estimation sample and provides a mapping between the index numbers that label the covariance parameters of the model and their associated values and labels for the alternative variable.

`estat covariance` computes the estimated variance–covariance matrix of the latent-variable errors for the alternatives. The estimates are displayed, and the variance–covariance matrix is stored in `r(cov)`.

estat correlation computes the estimated correlation matrix of the latent-variable errors for the alternatives. The estimates are displayed, and the correlation matrix is stored in r(cor).

estat facweights displays the covariance factor weights matrix and stores it in r(C).

estat mfx computes the simulated probability marginal effects.

Syntax for predict

predict [*type*] *newvar* [*if*] [*in*] [, *statistic* altwise]

predict [*type*] { *stub** | *newvarlist* } [*if*] [*in*], scores

statistic	Description
Main	
pr	probability alternative is chosen; the default
xb	linear prediction
stdp	standard error of the linear prediction

These statistics are available both in and out of sample; type predict ... if e(sample) ... if wanted only for the estimation sample.

Menu for predict

Statistics > Postestimation > Predictions, residuals, etc.

Options for predict

Main

pr, the default, calculates the probability that alternative j is chosen in case i.

xb calculates the linear prediction $\mathbf{x}_{ij}\boldsymbol{\beta} + \mathbf{z}_i\boldsymbol{\alpha}_j$ for alternative j and case i.

stdp calculates the standard error of the linear predictor.

altwise specifies that alternativewise deletion be used when marking out observations due to missing values in your variables. The default is to use casewise deletion. The xb and stdp options always use alternativewise deletion.

scores calculates the scores for each coefficient in e(b). This option requires a new variable list of length equal to the number of columns in e(b). Otherwise, use the *stub** option to have predict generate enumerated variables with prefix *stub*.

Syntax for estat

Alternative summary statistics

`estat alternatives`

Covariance matrix of the latent-variable errors for the alternatives

`estat covariance` [, `format(`*%fmt*`)` `border(`*bspec*`)` `left(#)`]

Correlation matrix of the latent-variable errors for the alternatives

`estat correlation` [, `format(`*%fmt*`)` `border(`*bspec*`)` `left(#)`]

Covariance factor weights matrix

`estat facweights` [, `format(`*%fmt*`)` `border(`*bspec*`)` `left(#)`]

Marginal effects

`estat mfx` [*if*] [*in*] [, *estat_mfx_options*]

estat_mfx_options	Description
Main	
`varlist(`*varlist*`)`	display marginal effects for *varlist*
`at(mean` [*atlist*] | `median` [*atlist*]`)`	calculate marginal effects at these values
Options	
`level(#)`	set confidence interval level; default is `level(95)`
`nodiscrete`	treat indicator variables as continuous
`noesample`	do not restrict calculation of means and medians to the estimation sample
`nowght`	ignore weights when calculating means and medians

Menu for estat

Statistics > Postestimation > Reports and statistics

Options for estat

Options for `estat` are presented under the following headings:

Options for estat covariance, estat correlation, and estat facweights
Options for estat mfx

Options for estat covariance, estat correlation, and estat facweights

`format(%fmt)` sets the matrix display format. The default for `estat covariance` and `estat facweights` is `format(%9.0g)`; the default for `estat correlation` is `format(%9.4f)`.

`border(bspec)` sets the matrix display border style. The default is `border(all)`. See [P] **matlist**.

`left(#)` sets the matrix display left indent. The default is `left(2)`. See [P] **matlist**.

Options for estat mfx

Main

`varlist(varlist)` specifies the variables for which to display marginal effects. The default is all variables.

`at(mean` [*atlist*] | `median` [*atlist*]`)` specifies the values at which the marginal effects are to be calculated. *atlist* is

[[*alternative*:*variable* = #] [*variable* = #] [...]]

The default is to calculate the marginal effects at the means of the independent variables at the estimation sample, `at(mean)`.

After specifying the summary statistic, you can specify a series of specific values for variables. You can specify values for alternative-specific variables by alternative, or you can specify one value for all alternatives. You can specify only one value for case-specific variables. For example, in `travel.dta`, `income` is a case-specific variable, whereas `termtime` and `travelcost` are alternative-specific variables. The following would be a legal syntax for `estat mfx`:

```
. estat mfx, at(mean air:termtime=50 travelcost=100 income=60)
```

When `nodiscrete` is not specified, `at(mean` [*atlist*]`)` or `at(median` [*atlist*]`)` has no effect on computing marginal effects for indicator variables, which are calculated as the discrete change in the simulated probability as the indicator variable changes from 0 to 1.

The mean and median computations respect any `if` and `in` qualifiers, so you can restrict the data over which the means or medians are computed. You can even restrict the values to a specific case; for example,

```
. estat mfx if case==21
```

Options

`level(#)` specifies the confidence level, as a percentage, for confidence intervals. The default is `level(95)` or as set by `set level`; see [U] **20.7 Specifying the width of confidence intervals**.

`nodiscrete` specifies that indicator variables be treated as continuous variables. An indicator variable is one that takes on the value 0 or 1 in the estimation sample. By default, the discrete change in the simulated probability is computed as the indicator variable changes from 0 to 1.

`noesample` specifies that the whole dataset be considered instead of only those marked in the `e(sample)` defined by the `asmprobit` command.

`nowght` specifies that weights be ignored when calculating the means or medians.

Remarks and examples

stata.com

Remarks are presented under the following headings:

Predicted probabilities

After fitting an alternative-specific multinomial probit model, you can use `predict` to obtain the simulated probabilities that an individual will choose each of the alternatives. When evaluating the multivariate normal probabilities via Monte Carlo simulation, `predict` uses the same method to generate the random sequence of numbers as the previous call to `asmprobit`. For example, if you specified `intmethod(Halton)` when fitting the model, `predict` also uses the Halton sequence.

▷ Example 1

In example 1 of [R] **asmprobit**, we fit a model of individuals' travel-mode choices. We can obtain the simulated probabilities that an individual chooses each alternative by using `predict`:

```
. use http://www.stata-press.com/data/r13/travel
. asmprobit choice travelcost termtime, case(id) alternatives(mode)
> casevars(income)
  (output omitted )
. predict prob
(option pr assumed; Pr(mode))
. list id mode prob choice in 1/12, sepby(id)
```

	id	mode	prob	choice
1.	1	air	.1494137	0
2.	1	train	.329167	0
3.	1	bus	.1320298	0
4.	1	car	.3898562	1
5.	2	air	.2565875	0
6.	2	train	.2761054	0
7.	2	bus	.0116135	0
8.	2	car	.4556921	1
9.	3	air	.2098406	0
10.	3	train	.1081824	0
11.	3	bus	.1671841	0
12.	3	car	.5147822	1

◁

Obtaining estimation statistics

Once you have fit a multinomial probit model, you can obtain the estimated variance or correlation matrices for the model alternatives by using the `estat` command.

▷ Example 2

To display the correlations of the errors in the latent-variable equations, we type

```
. estat correlation
```

	train	bus	car
train	1.0000		
bus	0.8909	1.0000	
car	0.7895	0.8951	1.0000

Note: correlations are for alternatives differenced with air

The covariance matrix can be displayed by typing

```
. estat covariance
```

	train	bus	car
train	2		
bus	1.600208	1.613068	
car	1.37471	1.399703	1.515884

Note: covariances are for alternatives differenced with air

◁

Obtaining marginal effects

The marginal effects are computed as the derivative of the simulated probability for an alternative with respect to an independent variable. A table of marginal effects is displayed for each alternative, with the table containing the marginal effect for each case-specific variable and the alternative for each alternative-specific variable.

By default, the marginal effects are computed at the means of each continuous independent variable over the estimation sample. For indicator variables, the difference in the simulated probability evaluated at 0 and 1 is computed by default. Indicator variables will be treated as continuous variables if the `nodiscrete` option is used.

▷ Example 3

Continuing with our model from example 1, we obtain the marginal effects for alternatives `air`, `train`, `bus`, and `car` evaluated at the mean values of each independent variable. Recall that the `travelcost` and `termtime` variables are alternative specific, taking on different values for each alternative, so they have a separate marginal effect for each alternative.

```
. estat mfx
```

Pr(choice = air) = .29434926

variable	dp/dx	Std. Err.	z	P>|z|	[95% C.I.	]	X
travelcost							
air	-.002688	.000677	-3.97	0.000	-.004015	-.001362	102.65
train	.0009	.000436	2.07	0.039	.000046	.001755	130.2
bus	.000376	.000271	1.39	0.166	-.000155	.000908	115.26
car	.001412	.00051	2.77	0.006	.000412	.002412	95.414
termtime							
air	-.010376	.002711	-3.83	0.000	-.015689	-.005063	61.01
train	.003475	.001639	2.12	0.034	.000264	.006687	35.69
bus	.001452	.001008	1.44	0.150	-.000523	.003427	41.657
car	.005449	.002164	2.52	0.012	.001209	.00969	0
casevars							
income	.003891	.001847	2.11	0.035	.000271	.007511	34.548

Pr(choice = train) = .29531182

variable	dp/dx	Std. Err.	z	P>|z|	[95% C.I.	]	X
travelcost							
air	.000899	.000436	2.06	0.039	.000045	.001753	102.65
train	-.004081	.001466	-2.78	0.005	-.006953	-.001208	130.2
bus	.001278	.00063	2.03	0.042	.000043	.002513	115.26
car	.001904	.000887	2.15	0.032	.000166	.003641	95.414
termtime							
air	.003469	.001638	2.12	0.034	.000258	.00668	61.01
train	-.01575	.00247	-6.38	0.000	-.020591	-.010909	35.69
bus	.004934	.001593	3.10	0.002	.001812	.008056	41.657
car	.007348	.002228	3.30	0.001	.00298	.011715	0
casevars							
income	-.00957	.002223	-4.31	0.000	-.013927	-.005214	34.548

Pr(choice = bus) = .08880039

variable	dp/dx	Std. Err.	z	P>|z|	[95% C.I.	]	X
travelcost							
air	.00038	.000274	1.39	0.165	-.000157	.000916	102.65
train	.001279	.00063	2.03	0.042	.000044	.002514	130.2
bus	-.003182	.001175	-2.71	0.007	-.005485	-.00088	115.26
car	.001523	.000675	2.26	0.024	.0002	.002847	95.414
termtime							
air	.001466	.001017	1.44	0.149	-.000526	.003459	61.01
train	.004937	.001591	3.10	0.002	.001819	.008055	35.69
bus	-.012283	.002804	-4.38	0.000	-.017778	-.006788	41.657
car	.00588	.002255	2.61	0.009	.001461	.010299	0
casevars							
income	.000435	.001461	0.30	0.766	-.002428	.003298	34.548

```
Pr(choice = car) = .32168607
```

variable	dp/dx	Std. Err.	z	P>|z|	[95% C.I.	]	X
travelcost							
air	.00141	.000509	2.77	0.006	.000411	.002408	102.65
train	.001903	.000886	2.15	0.032	.000166	.003641	130.2
bus	.001523	.000675	2.25	0.024	.000199	.002847	115.26
car	-.004836	.001539	-3.14	0.002	-.007853	-.001819	95.414
termtime							
air	.005441	.002161	2.52	0.012	.001205	.009677	61.01
train	.007346	.002228	3.30	0.001	.00298	.011713	35.69
bus	.005879	.002256	2.61	0.009	.001456	.010301	41.657
car	-.018666	.003938	-4.74	0.000	-.026385	-.010948	0
casevars							
income	.005246	.002166	2.42	0.015	.001002	.00949	34.548

First, we note that there is a separate marginal effects table for each alternative and that table begins by reporting the overall probability of choosing the alternative, for example, 0.2944 for air travel. We see in the first table that a unit increase in terminal time for air travel from 61.01 minutes will result in a decrease in probability of choosing air travel (when the probability is evaluated at the mean of all variables) by approximately 0.01, with a 95% confidence interval of about −0.016 to −0.005. Travel cost has a less negative effect of choosing air travel (at the average cost of 102.65). Alternatively, an increase in terminal time and travel cost for train, bus, or car from these mean values will increase the chance for air travel to be chosen. Also, with an increase in income from 34.5, it would appear that an individual would be more likely to choose air or automobile travel over bus or train. (While the marginal effect for bus travel is positive, it is not significant.)

◁

▷ Example 4

Plotting the simulated probability marginal effect evaluated over a range of values for an independent variable may be more revealing than a table of values. Below are the commands for generating the simulated probability marginal effect of air travel for increasing air travel terminal time. We fix all other independent variables at their medians.

```
. qui gen meff = .
. qui gen tt = .
. qui gen lb = .
. qui gen ub = .
. forvalues i=0/19 {
  2.          local termtime = 5+5*‘i’
  3.          qui replace tt = ‘termtime’ if _n == ‘i’+1
  4.          qui estat mfx, at(median air:termtime=‘termtime’) var(termtime)
  5.          mat air = r(air)
  6.          qui replace meff = air[1,1] if _n == ‘i’+1
  7.          qui replace lb = air[1,5] if _n == ‘i’+1
  8.          qui replace ub = air[1,6] if _n == ‘i’+1
  9.          qui replace prob = r(pr_air) if _n == ‘i’+1
 10. }
. label variable tt "terminal time"
```

```
. twoway (rarea lb ub tt, pstyle(ci)) (line meff tt, lpattern(solid)), name(meff)
> legend(off) title("  marginal effect of air travel" "terminal time and"
> "95% confidence interval", position(3))
. twoway line prob tt, name(prob) title("  probability of choosing" "air travel",
> position(3)) graphregion(margin(r+9)) ytitle("") xtitle("")
. graph combine prob meff, cols(1) graphregion(margin(l+5 r+5))
```



From the graphs, we see that the simulated probability of choosing air travel decreases in an sigmoid fashion. The marginal effects display the rate of change in the simulated probability as a function of the air travel terminal time. The rate of change in the probability of choosing air travel decreases until the air travel terminal time reaches about 45; thereafter, it increases.

◁

Stored results

`estat mfx` stores the following in `r()`:

Scalars

`r(pr_`*alt*`)` scalars containing the computed probability of each alternative evaluated at the value that is labeled X in the table output. Here *alt* are the labels in the macro `e(alteqs)`.

Matrices

`r(`*alt*`)` matrices containing the computed marginal effects and associated statistics. There is one matrix for each alternative, where *alt* are the labels in the macro `e(alteqs)`. Column 1 of each matrix contains the marginal effects; column 2, their standard errors; columns 3 and 4, their z statistics and the p-values for the z statistics; and columns 5 and 6, the confidence intervals. Column 7 contains the values of the independent variables used to compute the probabilities `r(pr_`*alt*`)`.

Methods and formulas

Marginal effects

The marginal effects are computed as the derivative of the simulated probability with respect to each independent variable. A set of marginal effects is computed for each alternative; thus, for J alternatives, there will be J tables. Moreover, the alternative-specific variables will have J entries, one for each alternative in each table. The details of computing the effects are different for alternative-specific variables and case-specific variables, as well as for continuous and indicator variables.

We use the latent-variable notation of `asmprobit` (see [R] **asmprobit**) for a J-alternative model and, for notational convenience, we will drop any subscripts involving observations. We then have the following linear functions $\eta_j = \mathbf{x}_j\boldsymbol{\beta} + \mathbf{z}\boldsymbol{\alpha}_j$, for $j = 1,\dots,J$. Let k index the alternative of interest, and then

$$\begin{aligned} v_{j'} &= \eta_j - \eta_k \\ &= (\mathbf{x}_j - \mathbf{x}_k)\boldsymbol{\beta} + \mathbf{z}(\boldsymbol{\alpha}_j - \boldsymbol{\alpha}_k) + \boldsymbol{\epsilon}_{j'} \end{aligned}$$

where $j' = j$ if $j < k$ and $j' = j - 1$ if $j > k$, so that $j' = 1,\dots,J-1$ and $\boldsymbol{\epsilon}_{j'} \sim \text{MVN}(\mathbf{0},\boldsymbol{\Sigma})$. Denote $p_k = \Pr(v_1 \le 0,\dots,v_{J-1} \le 0)$ as the simulated probability of choosing alternative k given profile $\mathbf{x}_k$ and $\mathbf{z}$. The marginal effects are then $\partial p_k/\partial \mathbf{x}_k$, $\partial p_k/\partial \mathbf{x}_j$, and $\partial p_k/\partial \mathbf{z}$, where $k = 1,\dots,J$, $j \ne k$. `asmprobit` analytically computes the first-order derivatives of the simulated probability with respect to the v's, and the marginal effects for $\mathbf{x}$'s and $\mathbf{z}$ are obtained via the chain rule. The standard errors for the marginal effects are computed using the delta method.

Also see

[R] **asmprobit** — Alternative-specific multinomial probit regression

[U] 20 Estimation and postestimation commands